Программирование арифметического сопроцессора


Все арифметические команды МП 8086 работают с целочисленными операндами. Но иногда возникает необходимость обрабатывать вещественные числа и использовать более сложные математические операции, чем сложение, вычитание, умножение и деление. Такие операции можно выполнять при использовании специального программного обеспечения, существенно замедляющего работу программы. Также для выполнения вычислений с использованием вещественных чисел к МП 8086 имеется возможность подключить специальный математический сопроцессор, который имеет большой набор различных математических операций и функций и достаточно быстро выполняет обработку вещественных чисел. 

Начиная с МП 80486DX сопроцессоры включаются в состав кристалла процессора. Для более ранних моделей процессоров сопроцессор вставляется в специальное гнездо на материнской плате. Название модели сопроцессора совпадает с названием модели соответствующего процессора, например сопроцессор 8087 соответствует процессору 8086, 80287 – 80286 и т. д. Сопроцессор обычно обозначается как FPU (Floating Point Unit) или как NPX (Numeric Processor eXtension).

Наличие сопроцессора позволяет значительно ускорить работу программ, выполняющих расчёты с высокой точностью, тригонометрические вычисления и обработку информации, которая должна быть представлена  в виде действительных чисел.

Определение наличия сопроцессора


Перед использованием сопроцессора необходимо определить его наличие в компьютере. Это актуально для процессоров 8086, 80286, 80386 и 80486 SX. В более старших моделей процессоров сопроцессор уже присутствует непосредственно в кристалле процессора. Существует несколько способов определения наличия сопроцессора. Каждый способ имеет свои достоинства и недостатки, а также определённую надёжность определения наличия сопроцессора. Под надёжностью определения наличия сопроцессора понимается вероятность того, что конкретный способ правильно определит его наличие. Далее рассмотрены способы определения наличия сопроцессора.

1. Чтение слова области данных BIOS. Это слово расположено в памяти по абсолютному адресу 0410H (сегмент 0040H и смещение 0010H). Если в прочтённом слове бит 1 установлен, то сопроцессор есть, если бит сброшен, сопроцессор отсутствует. Как показывает практика, такой способ определения наличия сопроцессора ненадёжен. Очевидно, это происходит из-за несоблюдения соглашений по содержанию информации в области данных BIOS его разработчиками. Рекомендуется пользоваться другими способами. Достоинством же данного способа является его простота. Пример определения наличия сопроцессора с использованием данного способа:

MOV AX, 0040H
MOV ES, AX


; настройка ES на область данных BIOS
TEST BYTE PTR ES:0010H, 0001H


; проверка 1-го бита

JZ None
. . .



; сопроцессор присутствует

None:

. . .



; сопроцессор отсутствует


2. Чтение информации о конфигурации компьютера. Этот способ зависит от типа ПК. В XT и некоторых моделях AT информация о конфигурации хранится в регистрах микросхемы 8255. Эта микросхема имеет три однобайтовых регистра A, B и C с номерами соответствующих портов 60H, 61H и 62H. Для того, чтобы считать содержимое из порта C, в котором хранится информация о конфигурации, необходимо предварительно установить бит 3 порта B. Если в считанном из порта C байте бит 1 установлен, то сопроцессор есть, в противном случае он отсутствует. Следующий пример показывает определение наличия сопроцессора с использованием данного способа.

IN AL, 61H


; чтение из порта B
OR AL, 00001000B

; установка 3-го бита

OUT 61H, AL


; запись в порт B
IN AL, 62H


; чтение из порта C
TEST AL, 00000010B

; проверка 1-го бита

JZ None
. . .



; сопроцессор присутствует

None:

. . .



; сопроцессор отсутствует

Достоинством способа является надёжность его работы в старых моделях компьютеров (XT), в отличии от первого способа. Недостатком способа является то, что на более старших моделях компьютеров он не всегда работает.


3. Вызов прерывания 11H, возвращающего информацию об оборудовании. Этот способ полностью аналогичен первому рассмотренному способу, так как прерывание 11H возвращает в регистре AX слово из области данных BIOS по адресу 0410H. Пример:

INT 11H



; вызов прерывания 11H
TEST AL, 00000010B

; проверка 1-го бита

JZ None
. . .



; сопроцессор присутствует

None:

. . .



; сопроцессор отсутствует


4. Чтение управляющего регистра сопроцессора. Этот способ основан на том, что если сопроцессор присутствует, то в его управляющем регистре будут установлены определённые биты. Для реализации этого способа нужно сначала попытаться инициализировать сопроцессор командой FINIT, а затем прочесть содержимое его управляющего регистра. При инициализации сопроцессора, если он присутствует, в управляющий регистр записывается определённый код. Для выделения из управляющего регистра определённых бит используется маска 0F3FH. Этой маской выделяются биты управления точностью результатов выполнения арифметических операций (с восьмого по одинадцатый) и биты управления маскированием исключительных ситуаций (с нулевого по пятый). Выделенное значение должно быть равно 033FH или, в двоичном представлении 0000001100111111B. Если это не так, то сопроцессора нет. Достоинством этого метода является возможность определения типа сопроцессора, так как разные сопроцессоры после инициализации возвращают отличающуюся информацию, например в 15-ом разряде регистра управления. Пример использования данного способа:

CrtlReg DW ?


; определение данных

. . .

FINIT



; инициализация сопроцессора

FSTCW CrtlReg


; чтение управляющего регистра 

AND CrtlReg, 0F3FH

; наложение маски

CMP CtrlReg, 033FH

; сравнение значений

JNZ None
. . .



; сопроцессор присутствует

None:

. . .



; сопроцессор отсутствует

Команды сопроцессора


Все команды сопроцессора имеют мнемонические обозначения, начинающиеся с буквы ‘F’. Коды этих команд состоят из специального префикса и непосредственно самого кода команды. При выполнении программы центральный процессор, встретив префикс команды сопроцессора, отдаёт саму команду на выполнение сопроцессору. Перед выполнением очередной команды сопроцессора необходимо выполнить команду центрального процессора WAIT, которая синхронизирует работу центрального процессора и сопроцессора. Команда WAIT не имеет операндов. Для процессоров, которые уже содержат сопроцессор, команду WAIT можно не использовать. При написании программы команду WAIT можно не ставить, так как транслятор Turbo Assembler автоматически вставляет её после каждой команды сопроцессора.

Операнды команд сопроцессора могут быть ячейками памяти или внутренними регистрами сопроцессора. Для передачи сопроцессору операндов из регистров центрального    процессора служит специальная команда центрального процессора ESC, которая будет рассмотрена далее в этой лекции. 

Архитектура сопроцессора с точки зрения программиста


Программисту доступны восемь регистров общего назначения сопроцессора, исмеющих мнемонические обозначения ST(0) – ST(7), и пять нечисловых регистров, которые будут рассмотренны позже. Десятибайтовые регистры ST(0) – ST(7) используются как стек, они предназначены для хранения операндов и результатов арифметических операций. Первый из этих регистров ST(0) можно обозначать просто как ST.


Структура стека сопроцессора представлена на рис. 1.

Рис. 1 – структура стека сопроцессора.

Из рисунка видно, что регистр ST(0) соответствует тому физическому регистру сопроцессора, который является вершиной стека, регистр ST(1) – следующему после вершины регистру и т. д. Таким образом, при обращении к регистру ST(0) происходит обращение к вершине стека сопроцессора. Направление роста стека определяет, что при записи в стек очередного значения, указатель стека уменьшается, а при выталкивании из стека очередного значения – увеличивается. При этом, если указатель стека доходит до 0, то при его уменьшении в нём устанавливается значение 7. Если же указатель стека доходит до 7, то при его увеличении в нём устанавливается значение 0. Таким образом, стек сопроцессора представляет собой как-бы кольцевой буфер.


При выполнении команд сопроцессора в стек (регистры ST(0) – ST(7)) заносятся операнды. После выполнения команды эти операнды выталкивыются из стека, а в стек заносится результат выполнения арифметической операции. Арифметические операции берут свои операнды с вершины стека и помещают результат в вершину стека. Рассмотрим, например, действие команды сложения вещественных чисел FADD. Эта команда имеет два операнда, которые берутся из вершины стека – из регистров ST(0) и ST(1). Результат выполнения команды FADD записывается в вершину стека в регистр ST(0). На рис. 2 показано содержимое стека до выполнения команды FADD, после занесения операндов в стек и после занесения в стек результата выполнения команды.

Рис. 2 – состояние стека в процессе выполнения команды FADD.

Существуют специальные команды сопроцессора, которые загружают регистры ST(0) – ST(7) значениями, взятыми из ячеек памяти. Другие команды наоборот извлекают из регистров ST(0) – ST(7) значения и записывают их в ячейки памяти. При выполнении этих команд автоматически выполняется преобразования типов операндов. Например, можно из ячейки памяти с целочисленным значением записать это значение в вершину стека и при этом будет выполнено преобразование целочисленного значения в вещественное. И наоборот, при записи значения из регистров ST(0) – ST(7) в память может выполняться преобразование вещественного значения в целочисленное.


Нечисловые регистры сопроцессора предназначены для хранения состояния сопроцессора и признаков значений регистров ST(0) – ST(7). Нечисловыми регистрами являются:

1. регистр управления;

2. регистр состояния;

3. регистр признаков;

4. указатель команды;

5. указатель операнда.

Структура нечисловых регистров сопроцессора представлена на рис. 3 (буквами ‘X’ обозначены неопределённые значения битов для некоторых моделей сопроцессоров).

Рис. 3 – нечисловые регистры сопроцессора.

Регистр управления


Биты 0 – 5 для всех сопроцессоров являются масками недействительных состояний. Если бит маски для какого-либо недействительного состояния, например, переполнения –  сброшен, то возникновение этого состояния вызывает прерывание центрального процессора. Если этот бит установлен, то прерывание не вырабатывается, а в качестве результата формируется особое значение (в рассматриваемом случае код бесконечности). Используются следующие маски особых случаев:


IM (Invalid Operation Mask) – маска недействительной операции;


DM (Denormalized Operand Mask) – маска денормализованного операнда;


ZM (Zero Divide Mask) – маска деления на нуль;


OM (Overflow Mask) – маска переполнения;


UM (Underflow Mask) – маска антипереполнения;


PM (Precision Mask) – маска особого случая при неточном результате.

В сопроцессорах 80287 и выше поля в битах 6 и 7 не используются. В сопроцессоре 8087 в поле 7 содержится маска недействительных прерываний IEM (Interrupt Enable Mask). Если IEM = 0, то прерывания центрального процессора не будет даже при возникновении незамаскированной ошибки.


Сопроцессор обычно использует вектор прерывания 13H. Процесс обработки прерывания 13H происходит так, как это было рассмотрено для аппаратных прерываний в лекции 8.


Содержимое поля PC (Precision Control, биты 8 и 9) определяет точность вычислений в сопроцессоре:


11B – используется расширенная точность;


10B – результат округляется до двойной точности;


00B – результат округляется до одинарной точности.


Двухбитовое поле RC (Rounding Control, биты 10 и 11) определяет режим округления при выполнении операций с вещественными числами:


00B – производится округление к ближайшему представимому числу. Этот режим устанавливается при инициализации сопроцессора;


01B – производится округление в направлении к отрицательной бесконечности;


10B – производится округление в направлении к положительной бесконечности;


11B – производится округление в направлении к нулю.


В сопроцессорах 8087 и 80287 содержимое бита 12 (IC – Infinity Control) предназначено для управления трактовкой понятия бесконечности. Данные сопроцессоры могут работать в двух режимах – проективном (IC = 0) и афинном (IC = 1). В проективном режиме существует только одна бесконечность, которая не имеет знака. В афинном режиме определено две бесконечности: положительная и отрицательная. В этом режиме допускается выполнение арифметических операций с бесконечностями. Сопроцессоры 80387 и выше работают только в режиме афинной арифметики.


После инициализации сопроцессора командой FINIT в регистре управления устанавливается режим работы с расширенной точностью и округления к ближайшему представимому числу. Все биты масок обработки особых случаев устанавливаются в 1, следовательно, все особые случаи будут замаскированы.

Регистр состояния


Биты 0 – 5 для являются признаками (флагами) особых случаев Они устанавливаются при возникновении следующих ошибок:


IE (Invalid Operation) – недействительная операция;


DE (Denormalized Operand) – денормализованный операнд;


ZE (Zero Divide) – деление на нуль;


OE (Overflow) – переполнение;


UE (Underflow Mask) – антипереполнение;


PE (Precision Mask) – потеря точности.

При возникновении прерывания от сопроцессора при помощи этих флагов можно определить, какой особый случай вызвал возникновение прерывания.


Бит 6 содержит флаг ошибки стека SF (Stack Fault). Этот флаг устанавливается сопроцессором при возникновении переполнения или антипереполнения стека. Переполнение стека возникает в том случае, если стек полностью заполнен при попытке записи в него очередного значения. Антипереполнение стека возникает в том случае, если при попытке вытолкнуть значение из стека он оказывается пустым. Флаг SF сбрасывается только при непосредственной записи значения в регистр состояния.


Бит 7 содержит флаг суммарной ошибки ES (Summary Error), который устанавливается при возникновении любого незамаскированного особого случая, т. е. установке любого из битов 0 – 5 регистра состояния. Это справедливо для сопроцессоров 80287 и выше. Для сопроцессора 8087 бит 7 содержит флаг IR (Interrupt Request) запроса прерывания при возникновении незамаскированного особого случая.


Биты C0, C1, C2 и C3 (Condition Code) – коды условий. Они определяются по выполнению команд сравнения и команд нахождения остатка.


В поле ST (Stack Pointer) содержится номер числового физического регистра, являющегося вершиной стека. Если стек пуст, ST = 0.


Бит занятости B (Busy) устанавливается, если сопроцессор выполняет команду или происходит прерывание от основного процессора. Если сопроцессор свободен, то бит занятости сбрасывается.


При инициализации сопроцессора все флаги, за исключением ST и ES (значения которых не определяются) сбрасываются.

Регистр признаков

Регистр признаков содержит восемь 2-х битных полей TAG0 – TAG7. Содержимое каждого поля характеризует значение соответствующего числового регистра ST(0) – ST(7). По содержимому полей TAG0 – TAG7 можно судить о том, какое число хранится в регистрах ST(0) – ST(7). Возможны следующие значения каждого поля TAG0 – TAG7:


00B – в числовом регистре находится действительное ненулевое число;


01B – в числовом регистре находится нуль;


10B – числовой регистр содержит недействительное число – бесконечность, специальное нечисловое значение NaN (Not a Number) или денормализованное число.


11B – числовой регистр пуст.

Если регистр не отмечен как пустой, то при попытке записи в него вырабатывается код недействительной операции (устанавливается бит 0 регистра состояния) и запись в стек не производится. Этот особый случай может быть замаскирован, и тогда запись будет произведена. Для этого нужно установить бит 0 регистра управления, иначе возникновение особого случая вызовет прерывание центрального процессора.

Указатель команды


Указатель команды содержит 20-разрядный физический адрес команды, вызвавшей особый случай, а также 11-разрядный код последней выполненной операции. Если при возникновении особого случая использовался операнд, то го адрес заносится в регистр указателя операнда.

Указатель операнда


Указатель операнда содержит 20-разрядный физический адрес операнда, использованного командой, вызвавшей возникновение особого случая.

Форматы вещественных чисел, используемые сопроцессором


Все числа в стеке сопроцессора представляются в действительном формате. Сопроцессор может использовать три формата вещественных чисел: с одинарной точностью, с двойной точностью и с расширенной точностью. Конкретный формат определяется содержимым поля PC (Precision Control) регистра управления (см. выше).


Каждый из форматов действительных чисел, используемых сопроцессором, содержит поле мантиссы, поле порядка и бит знака. Размер мантиссы и порядка определяется типом формата действительного числа и влияет на точность представления чисел. Чем больше разрядов содержит мантисса, тем точнее представляется число. Чем больше разрядов содержит порядок, тем шире диапазон представляемых чисел.


Форматы действительных чисел представлены на рис. 4.

Рис. 4 – форматы действительных чисел.

Следует отметить, что ухудшение точности не приводит к ускорению работы программы. Поэтому различные форматы следует использовать тогда, когда это целесообразно.


Все действительные числа, в независимости от формата, хранятся в нормализованном виде. Это означает, что мантисса числа имеет значения от 1 до 2. При этом целая часть числа, равная 1, не хранится, так как она заранее известна. Порядок числа определяет степень числа, на которое домножается мантисса. При этом к этой степени добавляется число 16383 для расширенной точности, 1023 для двойной точности и 127 для одинарной точности. Добавление константы к порядку нужно для того, чтобы полученное значение всегда было положительным, так как сам по себе порядок может быть и отрицательным.


При определении в программе вещественных чисел можно использовать директивы определения данных DD и DQ. При инициализации определённых таким образом переменных нужно записывать вещественное число в формате с плавающей точкой в десятичной форме, например 123.81263. Даже если число не имеет дробной части, его всё равно следует записать с десятичной точкой, например 5.0, иначе число будет интерпретированно транслятором как целое, а не вещественное.

Некоторые команды сопроцессора


Далее будут рассмотрены наиболее часто используемые команды сопроцессора 8087.

Команды передачи данных и загрузки констант сопроцессора


Эти команды позволяют загружать операнды из памяти в стек сопроцессора или извлекать из стека значения и записывать их в память. Некоторые команды загружают в стек сопроцессора константы, такие как 1 или число Пи. 


В таблице 1 приведены команды передачи данных и загрузки констант сопроцессора, их операнды и назначение.

Таблица 1 - команды передачи данных и загрузки констант сопроцессора.

	Команда
	Операнд
	Назначение

	FLD
	память
	Загрузка вещественного числа в ST(0)

	FILD
	память
	Загрузка целого числа в ST(0)

	FLDZ
	
	Загрузка нуля в ST(0)

	FLD1
	
	Загрузка единицы в ST(0)

	FLDPI
	
	Загрузка числа Пи в ST(0)

	FLDL2T
	
	Загрузка двоичного логарифма числа 10 в ST(0)

	FLDL2E
	
	Загрузка двоичного логарифма числа e в ST(0)

	FLDLG2
	
	Загрузка десятичного логарифма числа 2 в ST(0)

	FLDLN2
	
	Загрузка натурального логарифма числа 2 в ST(0)

	FST
	память
	Запись вещественного числа в память из ST(0) без выталкивания из стека

	FSTP
	память
	Запись вещественного числа в память из ST(0) с выталкиванием из стека

	FIST
	память
	Запись целого числа в память из ST(0) без выталкивания из стека

	FISTP
	память
	Запись целого числа в память из ST(0) с выталкиванием из стека


При использовании в качестве операнда ячейки памяти и косвенной формы адресации необходимо указывать тип операнда – конструкция WORD PTR для 16-битного целого числа, DWORD PTR для одинарной точности или для 32-битного целого числа, QWORD PTR для двойной точности или 64-битного целого числа, а также TBYTE PTR для формата с расширенной точностью.

Примеры использования команд:

A DW 1.0

B DD 2.0

C DQ 3.0

D DT 4.0

. . .

FLD B

MOV BX, OFFSET C

FLD QWORD PTR [BX]

MOV DI, OFFSET D

FLD TBYTE PTR [DI]

FILD A

MOV SI, OFFSET B

FILD DWORD PTR B

FILD C

. . .

FLDZ

FLD1

FLDPI

. . .

FSTP B

MOV BX, OFFSET C

FST QWORD PTR [BX]

MOV DI, OFFSET D

FSTP TBYTE PTR [DI]

FIST A

MOV SI, OFFSET B

FISTP DWORD PTR B

FISTP C
Арифметические команды сопроцессора


Арифметические команды сопроцессора выполняют такие действия с вещественными числами, как сложение, вычистание, умножение, деление, извлечение квадратного корня и другие. Далее рассмотрены наиболее часто используемые арифметические команды сопроцессора. 

Арифметические команды сложения, вычитания, умножения и деления имеют две формы: с выталкиванием из стека и без выталкивания. Команды с выталкиванием из стека (мнемоническое обозначение заканчивается на букву ‘P’) выталкивают из стека значение операнда после записи результата в стек. Команды без выталкивания из стека не выталкивают из стека значений операндов. 

Команды без выталкивания из стека, в свою очередь, также имеют две формы: выполнение операции над двумя регистрами и выполнение операции над регистром ST(0) и ячейкой памяти. В первом случае синтаксис команды следующий:

<команда> ST, <регистр> 

или

<команда> <регистр>, ST
В указанный регистр или регистр ST(0) помещается результат выполнения операции над двумя регистрами. Например:

<команда> ST, ST(1)

В этом примере результат выполнения операции над регистрами ST(0) и ST(1) помещается в регистр ST(0). Во втором случае синтаксис команды такой:

<команда> <операнд>

Операнд может быть ячейкой памяти размером 4 байта для одинарной точности или 8 байт для двойной точности.


Команды с выталкиванием из стека имеют следующий синтаксис:

<команда> <регистр>, ST
Сначала производится арифметическая операция, результат выполнения операции записывается в указанный регистр, а затем из стека выталкивается значение операнда (регистр ST(0)). Пример:

<команда> ST(2), ST
В этом примере производится арифметическая операция над регистрами ST(0) и ST(2), результат записывается в регистр ST(2), а затем из стека выталкивается значение операнда (регистр ST(0)). 


В таблице 2 представлены арифметические команды сопроцессора.

Таблица 2 – арифметические команды сопроцессора.

	Арифметическая операция
	Формы команды
	Синтаксис
	Выполняемые 

действия

	Сложение
	FADD
	FADD <ST>, <регистр>
	ST := ST + регистр

	
	
	FADD <регистр>, ST
	регистр := регистр + ST

	
	
	FADD <операнд>
	ST := ST + операнд

	
	FADDP
	FADDP <регистр>, ST
	регистр := регистр + ST,

выталкивание из стека

	Вычитание
	FSUB
	FSUB <ST>, <регистр>
	ST := ST – регистр

	
	
	FSUB <регистр>, ST
	регистр := регистр – ST

	
	
	FSUB <операнд>
	ST := ST – операнд

	
	FSUBP
	FSUBP <регистр>, ST
	регистр := регистр – ST,

выталкивание из стека

	Умножение
	FMUL
	FMUL <ST>, <регистр>
	ST := ST * регистр

	
	
	FMUL <регистр>, ST
	регистр := регистр * ST

	
	
	FMUL <операнд>
	ST := ST * операнд

	
	FMULP
	FMULP <регистр>, ST
	регистр := регистр * ST,

выталкивание из стека

	Деление
	FDIV
	FDIV <ST>, <регистр>
	ST := ST / регистр

	
	
	FDIV <регистр>, ST
	регистр := регистр / ST

	
	
	FDIV <операнд>
	ST := ST / операнд

	
	FDIVP
	FDIVP <регистр>, ST
	регистр := регистр / ST,

выталкивание из стека



Примеры использования команд:

X DD 0.1


; определение числа X
Y DD –0.5


; определение числа Y
. . .

FLD X


; загрузка числа X
FLD Y


; загрузка числа Y
FADDP ST(1), ST
; ST(1) := ST(1) + ST(0), 

; выталкивание операнда

. . .

FLD X


; загрузка числа X
FLD Y


; загрузка числа Y
FSUB ST, ST(1)

; ST(0) := ST(0) – ST(1)

. . .

FLD X


; загрузка числа X
FLD Y


; загрузка числа Y
FMUL ST(1), ST

; ST(1) := ST(1) * ST(0)

. . .

FLD X


; загрузка числа X
FDIV Y


; ST(0) := ST(0) / Y

Команда FSQRT

Команда FSQRT извлекает квадратный корень из значения регистра ST(0) и помещает результат в регистр ST(0). Пример использования команды FSQRT:

X DQ 225.0
; определение числа X
. . .

FLD X

; загрузка числа X
FSQRT

; извлечение корня

Команда FSCALE


Команда FSCALE умножает значение регистра ST(0) на двойку в степени значения регистра ST(1) и записывает результат в регистр ST(0), или ST(0) := ST(0) * 2ST(1). Значение степени числа 2 округляется до наименьшего целого числа.

Пример использования команды FSCALE:

X DQ 1.25

; определение числа X
Y DQ 16.0

; определение числа Y
. . .

FLD Y

; загрузка числа Y
FLD X

; загрузка числа X
FSCALE

; ST(0) := X * 2^Y

Команда FABS

Команда FABS помещает в регистр ST(0) абсолютное значение регистра ST(0), или ST(0) := |ST(0)|. Пример использования команды FABS:

X DQ –1.0

; определение числа X
. . .

FLD X

; загрузка числа X
FABS


; вычисление абсолютного значения

Команда FCHS

Команда FCHS изменяет знак числа в регистре ST(0) на противоположный и помещает результат в регистр ST(0), или ST(0) := –ST(0). Пример использования команды FCHS:

X DQ 100.0
; определение числа X
. . .

FLD X

; загрузка числа X
FCHS


; изменение знака

Команды трансцендентных функций


Команды трансцендентных функций позволяют вычислять значения таких функций как тангенс, арктангенс и некоторых других. Все команды трансцендентных функций требуют в качестве операндов регистры стека сопроцессора. Результаты вычислений также заносятся в регистры стека сопроцессора.

Команда FPATAN

Команда FPATAN вычисляет значение арктангенса. При этом вычисляется значение арктангенса частного значений регистров ST(1) и ST(0). Результат заносится в регистр ST(1), после чего из стека выталкивается операнд. Таким образом, после выполнения команды FPATAN в регистре ST(0) оказывается вычисленное значение арктангенса. Результат выполнения команды FPATAN имеет знак операнда в регистре ST(1) и по модулю не превосходит числа Пи. Операнды команды должны по абсолютному значению не превышать бесконечность.


Пример использования команды FPATAN:

X DQ 2.0

; определение числа X
Y DQ 1.0

; определение числа Y
. . .

FLD X

; загрузка числа X
FLD Y

; загрузка числа Y
FPATAN

; ST(0) := atan(X / Y)

Команда FPTAN

Команда FPTAN вычисляет значение тангенса. Операндом является регистр ST(0). Результат вычисления тангенса записывается в регистр ST(0) и в стек заносится 1. Таким образом, после выполнения команды FPTAN в регистре ST(0) оказывается 1, а в регистре ST(1) – вычисленное значение тангенса. Операнд команды должен быть в пределах от –263 до +263.


Пример использования команды FPTAN:

X DD 1.0

; определение числа X
. . .

FLD X

; загрузка числа X
FPTAN

; ST(0) := 1, ST(1) := tan(X)

Команда F2XM1


Команда F2XM1 вычисляет значение функции 2x – 1. Операндом является регистр ST(0). Результат вычисления значения функции заносится в регистр ST(0). Значение операнда должно быть в пределах от –1.0 до +1.0, в противном случае результат будет неопределённым.


Пример использования команды F2XM1:

X DQ 0.5

; определение числа X
. . .

FLD X

; загрузка числа X
F2XM1

; ST(0) := 2X-1

Команда  FYL2X

Команда  FYL2X вычисляет значение функции y * log2(x). Операндом y является регистр ST(1), а операндом x – ST(0). Результат вычисления значения функции заносится в регистр ST(1), после чего из стека выталкивается операнд. Таким образом, после выполнения команды FYL2X в регистре ST(0) оказывается вычисленное значение функции. Операнд ST(0) должен быть положительным ненулевым числом.


Пример использования команды FYL2X:

X DQ 4.0

; определение числа X
Y DQ 2.0

; определение числа Y
. . .

FLD X

; загрузка числа X
FLD Y

; загрузка числа Y
FYL2X

; ST(0) := Y * log2(X)

Команда FYLPXP1

Команда  FYL2XP1 вычисляет значение функции y * log2(x + 1). Операндом y является регистр ST(1), а операндом x – ST(0). Результат вычисления значения функции заносится в регистр ST(1), после чего из стека выталкивается операнд. Таким образом, после выполнения команды FYL2X P1 в регистре ST(0) оказывается вычисленное значение функции. Операнд ST(0) должен быть положительным ненулевым числом.


Пример использования команды FYL2X P1:

X DD 32.0

; определение числа X
Y DD 0.5

; определение числа Y
. . .

FLD X

; загрузка числа X
FLD Y

; загрузка числа Y
FYL2X

; ST(0) := Y * log2(X + 1)

Команды сравнения данных


Команды сравнения данных позволяют сравнивать между собой значения регистров стека сопроцессора и операндов из памяти. Результаты сравнения определяются флагами условий C0, C2 и C3. Все команды сравнения выполняют сравнение значений регистра ST(0) и какого-либо другого регистра или ячейки памяти. Значения флагов условий C0, C2 и C3 в зависимости от результатов сравнения представлены в таблице 3.

Таблица 3 – интерпретация значений флагов условий.

	Условие
	C0
	C2
	C3

	ST(0) > SRC
	0
	0
	0

	ST(0) < SRC
	0
	0
	1

	ST(0) = SRC
	1
	0
	0

	Особый случай
	1
	1
	1


SRC – это второй операнд, с которым производится сравнение. Это может быть регистр стека сопроцессора или ячейка памяти. Особым случаем является сравнение операндов, один из которых является нечисловым значением NaN или не может быть сравнён с другим операндом. Если флаг IM сброшен, то флаги условий при возникновении особого случая не устанавливаются.


В таблице 4 представлены команды сравнения данных сопроцессора.

Таблица 4 – команды сравнения данных сопроцессора.

	Команда сравнения
	Операнд
	Назначение

	FCOM
	
	сравнение ST(0) и ST(1)

	FCOM
	память
	сравнение ST(0) и значения ячейки памяти

	FCOMP
	память
	сравнение ST(0) и значения ячейки памяти, выталкивание из стека операнда

	FCOMPP
	
	сравнение ST(0) и ST(1) и выталкивание из стека двух операндов

	FTST
	
	сравнение ST(0) с 0.0



Примеры использования команд сравнения данных сопроцессора:

X DD 1.0

; определение числа X
Y DD 1.01

; определение числа Y
. . .

FLD X

; загрузка числа X
FLD Y

; загрузка числа Y
FCOM


; сравнение ST(0) и ST(1)

. . .

FLD X

; загрузка числа X
FCOMP Y

; сравнение ST(0) и Y
. . .

FLD Y

; загрузка числа Y
FTST


; сравнение Y с 0.0

. . .

FLD X

; загрузка числа X
FLD Y

; загрузка числа Y
FCOMPP

; сравнение X и Y
Команды управления сопроцессором


Команды управления сопроцессором позволяют выполнять инициализацию сопроцессора, сохранение значения регистра состояния и загрузку регистра управления сопроцессора, а также выполнять сохранение значений всех регистров сопроцессора в памяти и загрузку этих значений из памяти.

Команда FINIT

Команда FINIT выполняет инициализацию сопроцессора. Инициализация включает в себя установку в начальное состояние всех регистров сопроцессора (числовых и нечисловых). Команда FINIT выполняет следующие действия:

1. В регистр управления сопроцессора заносится значение 037FH (округление до ближайшего представимого числа, расширенная точность, все особые случаи замаскированы).

2. В регистр состояния заносится значение 0000H (флаги особых случаев сброшены, вершина стека установлена в 0).

3. Все регистры стека сопроцессора помечаются в регистре признаков как пустые. Сами значения регистров стека не изменяются.

4. В регистр указателя команды заносится значение 0000H.

5. В регистр указателя операнда заносится значение 0000H.


Команду FINIT рекомендуется выполнить перед началом работы программы с сопроцессором для возвращения его в исходное состояние после возможного использования другими программами.

Команда FSTSW

Команда FSTSW сохраняет значение регистра состояния сопроцессора в 2-х байтовой ячейке памяти. Синтаксис команды:

FSTSW <адресное_выражение>

В качестве адресного выражения может использоваться прямая и косвенная адресация памяти. Тип операнда можно не указывать, так как он известен заранее.


Примеры использования команды FSTSW:

SW DW ?


; переменная для сохранения значения

. . .

MOV BX, OFFSET SW
; настройка BX на SW
FSTSW [BX]

; сохранение значения регистра состояния

Команда FLDCW

Команда FLDCW загружает в регистр управления сопроцессора слово из памяти. Синтаксис команды:

FLDCW <адресное_выражение>

В качестве адресного выражения может использоваться прямая и косвенная адресация памяти. Тип операнда можно не указывать, так как он известен заранее. Команда FLDCW вместе с командой FSTCW может использоваться для установки необходимых режимов работы сопроцессора (точность, управление округлением, маски особых случаев).


Пример использования команды FLDCW:

CW DW ?

; переменная для загрузки регистра управления

. . .

FLDCW CW

; загрузка регистра управления сопроцессора

Команда FSTCW

Команда FSTCW сохраняет значение регистра управления сопроцессора в ячейке памяти. Синтаксис команды:

FSTCW <адресное_выражение>


В качестве адресного выражения может использоваться прямая и косвенная адресация памяти. Тип операнда можно не указывать, так как он известен заранее. Команда FSTCW может использоваться для последующей установки необходимых режимов работы сопроцессора совместно с командой FLDCW.


Пример использования команды FSTCW:

CW DW ?

; переменная для сохранения регистра управления

. . .

FSTCW CW

; сохранение регистра управления сопроцессора


Пример использования команд FLDCW и FSTCW для установки двойной точности и округления в сторону нуля:

CW DW ? 

; переменная для сохранения регистра управления

. . .

FSTCW CW



; сохранение регистра управления

OR CW, 0000111000000000B
; установка нужного режима работы

FLDCW CW



; загрузка регистра управления

Команда FCLEX

Команда FCLEX сбрасывает в регистре состояния сопроцессора флаги особых случаев (PE, OE, UE, ZE, DE и IE), флаг суммарной ошибки ES, флаг ошибки стека SF и флаг занятости B.

Команда FSAVE

Команда FSAVE сохраняет в памяти значения всех числовых и нечисловых регистров сопроцессора и затем повторно инициализирует сопроцессор. Формат, в котором команда FSAVE записывает значения нечисловых регистров в память, представлен на рис. 5.

	15
	0
	

	Регистр управления
	0

	Регистр состояния
	2

	Регистр признаков
	4

	Указатель команды (биты 0 – 15)
	6

	Указатель команды (биты 16 – 31)
	8

	Указатель операнда (биты 0 – 15)
	10

	Указатель операнда (биты 16 – 31)
	12


Рис. 5 – формат записи нечисловых регистров в памяти командой FSAVE.

Сразу за значениями указанных регистров в памяти записываются восемь 10-байтовых значений, определяющих содержимое восьми регистров стека сопроцессора. Синтаксис команды FSAVE:

FSAVE <адресное_выражение>

В качестве адресного выражения может использоваться прямая и косвенная адресация памяти. При использовании косвенной формы адресации тип операнда не указывается. Начиная со смещения, определяемого адресным выражением, в память будут записаны значения всех регистров сопроцессора.


Пример использования команды FSAVE:

S DB (14 + 10 * 8) DUP (?)
; область памяти 

; для сохранения регистров

. . .

FSAVE S

; сохранение значений регистров сопроцессора

Команда FRSTORE

Команда FRSTORE заносит в регистры сопроцессора значения, ранее сохранённые командой FSAVE. Синтаксис команды FRSTORE – такой же, как и у команды FSAVE.


Пример использования команды FRSTORE:

S DB (14 + 10 * 8) DUP (?)
; область памяти 

; для сохранения регистров

. . .

FSAVE S

; сохранение значений регистров сопроцессора

FRSTORE S

; восстановление значений регистров сопроцессора

Команда FFREE

Команда FFREE помечает регистр стека сопроцессора как свободный. Эта команда заносит код 11B в регистр признаков сопроцессора для указанного регистра стека. Синтаксис команды:

FFREE ST(i)

i – целое число от 0 до 7, определяющее необходимый регистр стека сопроцессора. При выполнении команды FFREE содержимое указанного регистра и значение указателя стека не изменяются.


Пример использования команды FFREE:

FFREE ST(0)

FFREE ST(1)

FFREE ST(2)

Команда FNOP


Команда FNOP для сопроцессора аналогична команде NOP для центрального процессора – она не выполняет никаких действий и может использоваться для отладки программ, написанных с использованием сопроцессора.

Примеры решения типовых вычислительных задач

1. Вычисление интеграла функции


В качестве первого примера рассмотрим приближённое вычисление интеграла от функции Sin(x) методом Симпсона по трём точкам на отрезке от /3 до 3/4. Интеграл вычисляется по следующей формуле:

где a и b – границы отрезка (в данном случае a = /3, b = 3/4), h – шаг (расстояние между узлами интегрирования: h = (b – a) / 2 = 5/24. Для вычисления интеграла сначала нужно составить таблицу значений функции Sin(x) для точек a, b и (a + 2) / 2:

	Sin(/3)
	0,866025403784438646763723170752936

	Sin(13/24)
	0,991444861373810411144557526928563

	Sin(3/4)
	0,707106781186547524400844362104849


Эти значения используются в программе для вычисления значения интеграла. Результат вычисления значения интеграла заносится в регистр ST(0). Далее представлен текст программы:

CODE SEGMENT

; сегмент команд

        ASSUME CS:CODE, DS:DATA, SS:STAC

Start:  MOV AX, DATA
; настройка DS
        MOV DS, AX
; на сегмент данных

        FINIT

; инициализация сопроцессора

        FLD n4

; загрузка в стек числа 4.0

        FMUL Sin[8]
; умножение на значение функции в точке (a + b) / 2

        FADD Sin[0]
; прибавление значения функции в точке a
        FADD Sin[16]
; прибавление значения функции в точке b
        FDIV n3

; деление на число 3.0

        FLD B

; вычисление

        FSUB A

; значения

        FDIV n2

; шага h
        FMUL ST, ST(1)
; умножение шага на полученное значение

        MOV AX, 4C00H
; функция завершения программы

        INT 21H

; вызов DOS

CODE ENDS

DATA SEGMENT

; сегмент данных
; определение границ отрезка интегрирования

        A DQ 1.04719755119659774615421446109317

        B DQ 2.35619449019234492884698253745963

; определение промежуточных чисел

        n2 DD 2.0

        n3 DD 3.0

        n4 DD 4.0

; определение таблицы значений функции Sin
        Sin DQ 0.866025403784438646763723170752936

            DQ 0.991444861373810411144557526928563

            DQ 0.707106781186547524400844362104849

DATA ENDS

STAC SEGMENT PARA STACK
; сегмент стека
        DB 128 DUP (?)
; область памяти под стек

STAC ENDS

     END Start

; точка входа
Для обращения к элементам таблицы значений функции используются смещения 8 (для второго элемента) и 16 (для третьего элемента), поскольку размер элемента таблицы – 8 байт.

2. Вычисление суммы степенного ряда


В качестве второго примера будет рассмотрена программа вычисления суммы степенного ряда для функции ex.


Степенной ряд для функции ex записывается в виде:

При вычислении значения функции ex будет использоваться машинная точность, т. е. максимально возможная точность для выбранного формата представления действительных чисел. В примере используется формат с расширенной точностью. Переменная X содержится в сегменте данных программы и имеет двойную точность (8 байт). Программа завершает работу, когда достигнут машинный нуль, т. е. когда при добавлении очередного слагаемого к сумме степенного ряда значение этой суммы не изменяется. В этом случае вычисленное значение функции имеет машинную точность. Далее представлен текст программы.

CODE SEGMENT

; сегмент команд

        ASSUME CS:CODE, DS:DATA, SS:STAC

Start:  MOV AX, DATA
; настройка DS
        MOV DS, AX
; на сегмент данных

        FINIT

; инициализация сопроцессора

        FLD1

; загрузка 1.0 в стек

        FLD1

; ---//---

        FLD1

; ---//---

        FLD1

; ---//---

        FLD X

; загрузка переменной X в стек

M1:     FADD ST(1), ST
; добавление к сумме ряда очередного слагаемого

        FXCH ST(2)
; обмен ST(0) и ST(2)

        FCOM

; сравнение значений предыдущей и текущей суммы

        FSTSW SW

; загрузка регистра состояния в переменную SW
        AND SW, 4500H
; наложение маски

        CMP SW, 4000H
; проверка на установку флага C3

        JE M2

; переход, если достигнут машинный нуль

        FXCH ST(2)
; обмен ST(0) и ST(2)

        FADD ST(2), ST
; получение предыдущей суммы ряда

        FMUL X

; умножение предыдущего слагаемого на X
        FXCH ST(3)
; обмен ST(0) и ST(3)

        FLD1

; загрузка 1.0 в стек

        FADDP ST(1), ST
; увеличение аргумента факториала

        FXCH ST(3)
; обмен ST(0) и ST(3)

        FDIV ST, ST(3)
; деление предыдущего слагаемого на арг. факториала

        JMP M1

; следующая итерация вычислений

M2:     MOV AX, 4C00H
; функция завершения программы

        INT 21H

; вызов DOS

CODE ENDS

DATA SEGMENT

; сегмент данных
        X DQ 1.0

; аргумент функции ex
        SW DW ?

; переменная для хранения регистра состояния

DATA ENDS

STAC SEGMENT PARA STACK
; сегмент стека
        DB 128 DUP (?)
; область памяти под стек

STAC ENDS

     END Start

; точка входа
Значение следующего слагаемого вычисляется из значения предыдущего слагаемого следующим образом. Предыдущее слагаемое домножается на значение переменной X и делится на аргумент факториала для последующего слагаемого. Аргумент факториала для последующего слагаемого вычисляется путём добавления единицы к аргументу факториала предыдущего слагаемого. После вычисления значения очередного слагаемого оно добавляется к результату. Затем сравниваются значения предыдущего и текущего значений суммы ряда. Если они равны, значит достигнут машинный нуль и программа завершает работу, при этом результат находится в регистре ST(0). Если они не равны, то вычисления продолжаются дальше. В качестве начального слагаемого в стек заносится две единицы – для предыдущего и текущего значений суммы ряда. Ещё две единицы заносятся в стек для начального аргумента факториала и для значения, на которое увеличивается аргумент факториала. Для получения значений флагов условий используется команда FSTSW и маска 4500H, выделяющая значения необходимых бит. При равенстве сравниваемых операндов флаг C3 должен быть установлен, а флаги C2 и C0 – сброшены. Это соответствует значению 4000H после наложения маски на значение регистра состояния сопроцессора.

3. Нахождение корня уравнения методом Ньютона


Третий пример демонстрирует программу, вычисляющую приближённое значение корня уравнения 2x2 – 4x + 1 = 0 на отрезке от 0 до 0.5 методом Ньютона.


Метод Ньютона позволяет получить приближённое значение корня уравнения, используя итерационный процесс, по следующей формуле:

где Xi+1 – следующее приближение (уточнённое значение корня), Xi - приближённое значение корня на предыдущем шаге. В качестве начального приближения выбирается та точка отрезка, в которой выполняется условие 

. В примере итерационный процесс выполняется до тех пор, пока не будет достигнут машинный нуль. Далее представлен текст программы.

CODE SEGMENT

; сегмент команд
        ASSUME CS:CODE, DS:DATA, SS:STAC

Start:  MOV AX, DATA
; настройка DS

        MOV DS, AX
; на сегмент данных

        FINIT

; инициализация сопроцессора

        FLDZ

; запись в стек начального приближения

        FLDZ

; выделение места для предыдущего приближения

M1:     FXCH ST(1)
; обмен ST(1) и ST(0)

        FADDP ST, ST
; выталкивание из стека значения

        FLDZ

; загрузка в стек нуля

        FADD ST, ST(1)
; дублирование предыдущего приближения

        FMUL ST, ST
; вычисление xi2
        FLD n2

; загрузка в стек числа 2.0

        FMULP ST(1), ST
; вычисление 2xi2
        FLD n4

; загрузка в стек числа 4.0

        FMUL ST, ST(2)
; вычисление 4xi
        FSUBP ST(1), ST
; вычисление 2xi2 - 4xi
        FLD1

; загрузка в стек единицы

        FADDP ST(1), ST
; вычисление 2xi2 - 4xi + 1

        FLD n4

; загрузка в стек числа 4.0

        FMUL ST, ST(2)
; вычисление 4xi 

        FLD n4

; загрузка в стек числа 4.0

        FSUBP ST(1), ST
; вычисление 4xi - 4

        FDIVP ST(1), ST
; вычисление (2xi2 - 4xi + 1) / (4xi – 4)

        FCHS

; вычисление -(2xi2 - 4xi + 1) / (4xi – 4)

        FADD ST, ST(1)
; вычисление xi - (2xi2 - 4xi + 1) / (4xi – 4)

        FCOM

; сравнение предыдущего и текущего приближений

        FSTSW SW

; загрузка регистра состояния в переменную SW
        AND SW, 4500H
; наложение маски

        CMP SW, 4000H
; проверка на установку флага C3

        JNE M1

; переход, если не достигнут машинный нуль

        MOV AX, 4C00H
; функция завершения программы

        INT 21H

; вызов DOS

CODE ENDS

DATA SEGMENT

; сегмент данных
        n1 DD 1.0

; определение промежуточных чисел

        n2 DD 2.0

        n4 DD 4.0

        SW DW ?

; переменная для хранения регистра состояния

DATA ENDS

STAC SEGMENT PARA STACK
; сегмент стека

        DB 128 DUP (?)
; область памяти под стек

STAC ENDS

     END Start

; точка входа

В начале программы командой FLDZ в стек сопроцессора загружается начальное приближение. На каждой итерации последовательно вычисляются значение функции в точке предыдущего приближения и значение производной функции. После нахождения этих значений вычисляется следующее приближение по формуле, приведённой выше. После нахождения следующего приближение оно сравнивается с предыдущим. Программа завершает работу, когда текущее и предыдущее приближения равны. Это свидельствует о достижении машинного нуля. Результат работы программы – приближённое значение корня уравнения – заносится в регистр ST(0).

