Понятие резидентной программы


Резидентная программа – это специальная программа, которая после своего завершения оставляет часть своего кода в памяти (резидентная часть) и этот код вызывается при возникновении какого-либо прерывания (аппаратного или программного). Резидентные программы – это единственное исключение, когда в операционной системе MS-DOS «одновременно» могут работать несколько программ. Примерами резидентных программ могут являться загружаемые русификаторы, различные утилиты, которые вызываются при нажатии какой-то определённой комбинации клавиш, хранители экрана, резидентные часы и т. д. 


Полноценная резидентная программа должна иметь средства проверки на повторную загрузку и возможность выгрузки из памяти. Средства проверки на повторную загрузку не дают программе загрузиться в память, если там уже присутствует её копия, загруженная раньше. Возможность выгрузки программы из памяти обеспечивает в случае необходимости, по команде оператора завершение работы резидентной части программы и выгрузку её из памяти.


Нерезидентная часть программы обычно выполняет проверку на повторную загрузку, обрабатывает ключи командной строки и оставляет часть кода резидентным в памяти. После этого нерезидентная часть завершается, а в памяти остаётся резидентная часть.


Резидентные программы обычно «перехватывают» вектора прерываний и при возникновении прерывания вместо старого обработчика, вектор которого был перехвачен, используется обработчик резидентной программы. Иногда бывает необходимо помимо выполнения собственного кода резидентной части вызвать и старый обработчик, который может выполнять какие-то стандартные действия для обработки этого прерывания. Резидентная же часть выполняет какие-либо дополнительные действия помимо стандартных. Поскольку резидентная программа должна по возможности обеспечивать выгрузку своего кода из памяти и вызов старого обработчика, перед тем, как «перехватить» вектор прерывания, т. е. перезаписать его новым значением, программа должна сохранить в своих ячейках памяти значение старого вектора прерывания.

Структура резидентной программы


Как уже было сказано, резидентная программа состоит из двух частей – резидентной и нерезидентной. Резидентная часть, в свою очередь, состоит по крайней мере из одной процедуры обработки прерывания. Обычно резидентная часть помимо процедуры обработки прерывания включает в себя и ячейки памяти в которых хранятся внутренние данные резидентной части или, например, старое значение вектора прерывания. На рис. 1 показана обобщённая структура резидентной программы для операционной системы MS-DOS.

Рис. 1 – обобщённая структура резидентной программы.

В зависимости от конкретных требований к резидентной программе некоторые элементы обобщённой структуры могут быть исключены из резидентной программы. Минимальная резидентная программа должна содержать хотя бы один обработчик прерывания и процедуру загрузки резидентной части. 


Реализация структуры резидентной программы во многом зависит от типа программы – COM или EXE. Например, для программ типа EXE стеки резидентной и нерезидентной частей можно оформить в виде отдельных программных сегментов, выравнянных по границе параграфа. Данные резидентной и нерезидентной частей также можно оформить в виде отдельных программных сегментов. Для программы типа COM все элементы обобщённой структуры резидентной программы должны находиться в одном программном сегменте. Это вызывает некоторые сложности при переключении стеков.

Далее будут подробно рассмотрены все элементы обобщённой структуры резидентной программы.

Стек резидентной части


В стеке резидентной части обычно сохраняются значения используемых регистров при входе в процедуры обработки прерываний (см. лекцию 8), а также промежуточные данные резидентной части. Если в стеке требуется сохранить только значения регистров то, в принципе, можно использовать и стек прерванной программы. Но для повышения надёжности работы резидентной программы и для обеспечения её независимости от других программ всё же рекомендуется создать свой собственный стек и выполнить переключение на него. Рассмотрим, как можно выполнить переключение стека для программы типа COM. Вся резидентная программа должна располагаться в одном программном сегменте. Для переключения на стек в COM программе потребуется настроить регистр SS на сегмент кода, т. е. на регистр CS, а в регистр SP занести смещение последнего байта, отведённого под стек. Пример:

CODE SEGMENT


. . .


MyProc PROC


MOV CS:OLD_AX, AX
; сохранение AX

MOV CS:OLD_SS, SS
; сохранение SS


MOV CS:OLD_SP, SP
; сохранение SP


MOV AX, CS


MOV SS, AX

; переключение стека


MOV SP, OFFSET Stac + 128



. . .


MOV AX, CS:OLD_SS


MOV SS, AX

; восстановление стека


MOV SP, CS:OLD_SP


MOV AX, CS:OLD_AX
; восстановление AX



IRET


MyProc ENDP

OLD_SS DW ?

OLD_SP DW ?

OLD_AX DW ?

Stac DB 128 DUP (?)


. . .

CODE ENDS

В программах типа COM создавать собственный стек для нерезидентной части не нужно.

Процедура загрузки резидентной части


Эта процедура является основной для нерезидентной части, любая резидентная программа должна содержать эту процедуру. Процедура загрузки резидентной части выполняет следующие действия:

1. сохраняет в переменных резидентной части значение вектора прерывания, который будет перехвачен;

2. заменяет значение перехватываемого вектора прерывания на новое;

3. оставляет резидентную часть программы в памяти и завершает программу.

Для сохранения значения перехватываемого вектора прерывания требуется вычислить смещение от начала таблицы векторов прерываний и прочитать по этому смещению старое значение вектора прерывания, а затем записать его в переменные резидентной части. Пример (номер вектора прерывания находится в регистре BX):

OLD_VEC DD ?
; здесь будет храниться старое значение




; вектора прерывания

. . .

SHL BX, 1

; вычисление смещения

SHL BX, 1

XOR AX, AX
; настройка ES на начало таблицы векторов
MOV ES, AX
; прерываний
MOV AX, ES:[BX]

MOV WORD PTR OLD_VEC, AX

MOV AX, ES:[BX + 2]

MOV WORD PTR OLD_VEC + 2, AX

Замена значения вектора прерывания подробно рассматривалась в лекции 8.

Для того, чтобы оставить резидентную часть программы в памяти служит специальная функция прерывания DOS 21H. Прерывание DOS – это вызов диспетчера функций, предоставляемых операционной системой MS-DOS пользовательским программам. При этом конкретная функция определяется номером, который записывается перед вызовом прерывания 21H в регистр AH. Функция может иметь параметры, которые записываются в другие регистры. Конкретные параметры и регистры, в которые они должны быть записаны зависят от самой функции. Для загрузки резидентной части в память служит функция DOS 31H, при этом для этой функции необходимо в регистре DS указать сегмент резидентной части, а в регистре DX – объём памяти, который занимает резидентная часть в параграфах (блоках по 16 байт). Функция DOS 31H помимо загрузки резидентной части завершает выполнение программы. 

Для вычисления размера резидентной части в программах типа COM можно в конце резидентной части поставить какую-либо метку, смещение которой и будет размером резидентной части в байтах. При этом резидентная часть должна располагаться сначала сегмента команд, тогда размер будет вычислен правильно. Если резидентная часть располагается не в начале сегмента команд, то в памяти останется не только сама резидентная часть, но и те коды или данные, что были перед ней в начале сегмента команд. Это приводит только к лишнему расходу памяти, в остальном программа будет работать корректно. Если резидентная часть использует переключение стека, то стек должен входить в резидентную часть. То же относится и к переменным резидентной части. После того как вычислен размер резидентной части в байтах, нужно разделить это значение на 16 (т. е. вычислить сколько это будет параграфов) и прибавить единицу, чтобы округлить размер в сторону большего параграфа. Это следует выполнить, так как если округлить размер в сторону меньшего параграфа, несколько байт резидентной части останутся незагруженными в память, что может привести к некорректной работе программы. Пример вычисления размера резидентной части в параграфах для программы типа COM:

CODE SEGMENT

Start:


OLD_VEC DD ?
; начало резидентной части

. . .


MyProc PROC



. . .



IRET


MyProc ENDP


. . .


Stac DB 128 DUP (?)


. . .

M1:



; конец резидентной части

. . .


MOV DX, (M1 – Start + 10FH) / 16

; теперь в DX размер резидентной части в параграфах

. . .

CODE ENDS


Разберём как практически вычисляется размер резидентной части в параграфах. Дело в том, что выражение виде (OFFSET M1) / 16 + 1 записать нельзя, так как на этапе трансляции Turbo Assembler ещё не знает значения выражения OFFSET M1. Для решения этой проблемы можно вычислить разность между двумя метками – началом и концом резидентной части, как это показано в примере. Однако COM программа имеет ещё и префикс программного сегмента, размер которого в 256 байт тоже должен быть учтён. Число 10FH – это размер префикса программного сегмента 100H плюс число 0FH, которое требуется для округления размера резидентной части в параграфах к наибольшему целому параграфу. Исходя из этих рассуждений выражение (M1 – Start + 10FH) / 16 и представляет собой размер резидентной части в параграфах.

Для программ типа EXE вычисление размера резидентной части может оказаться сложнее. Прежде всего следует разместить сегмент стека резидентной части, сегмент данных резидентной и сегмент команд резидентной части в одном месте в начале программы, т. е. эти три сегмента должны идти последовательно, но всё равно в каком порядке. Сегмент стека или сегмент данных могут отсутствовать. Если и резидентная и нерезидентная части программы расположены в одном программном сегменте, то вычисление размера резидентной части в параграфах аналогично случаю для COM программы. Рассмотрим вычисление размера резидентной части, если она состоит из трёх сегментов – сегмента команд, сегмента данных и сегмента стека. Первая возможная ситуация – когда сегмент команд является общим для резидентной и нерезидентной части. В этом случае для вычисления размера всей резидентной части, включая её сегменты данных и стека, необходимо сначала вычислить размер сегментов данных и стека, а затем прибавить к этому значению размер кода резидентной части. В следующем примере показано, как это можно реализовать:

Data SEGMENT


. . .

Data ENDS

Stac SEGMENT PARA


DB 128 DUP (?)

Stac ENDS

Code SEGMENT

Start:


. . .

; код резидентной части
M1:


. . .

; код нерезидентной части

MOV DX, Code 

SUB DX, Data

ADD DX, (M1 – Start + 10FH) / 16

; теперь в DX размер резидентной части в параграфах

. . .

Code ENDS

Вторая возможная ситуация – когда сегменты команд отдельны для резидентной и нерезидентной частей и сегмент команд нерезидентной части следует сразу за сегментом команд резидентной части. При этом сегмент команд нерезидентной части должен иметь параметр «PARA», обозначающий выравнивание содержимого сегмента по границе параграфа. Размер резидентной части в параграфах в этом случае – это разница значений сегментов начала сегмента команд нерезидентной части и начального сегмента резидентой части. Пример:

Data SEGMENT


. . .

Data ENDS

Stac SEGMENT PARA


DB 128 DUP (?)

Stac ENDS

Resident SEGMENT


. . .

; код резидентной части
Resident ENDS

Code SEGMENT PARA


. . .

; код нерезидентной части

MOV DX, Code


SUB DX, Data


ADD DX, 10H

; теперь в DX размер резидентной части в параграфах

. . .

Code ENDS


Дополнительные 10H параграфов прибавляются для учёта размера префикса программного сегмента.


Теперь, когда вычислен размер резидентной части нужно занести его в регистр DX, а в регистр AL занести код возврата из программы (обычно это просто 0). В регистр AH заносится номер функции 31H и вызывается прерывание DOS 21H. Резидентная часть программы остаётся в памяти, а сама программа завершается и управление передаётся операционной системе. 
Примеры простейших резидентных программы типа COM

Следующий пример демонстрирует полностью законченную резидентную программу, которая перехватывает прерывание от клавиатуры (вектор прерывания для клавиатуры под номером 09H) и в программе обработки прерывания вызывает старый обработчик прерывания от клавиатуры. Это программа типа COM.

CODE SEGMENT


ASSUME CS:CODE, DS:CODE


ORG 100H

Start:


; начало резидентной части


JMP Init


OLD_VEC DD ?
; значение старого вектора

KbProc PROC

; процедура обработки прерывания


PUSHF


CALL CS:OLD_VEC
; вызов старого обработчика


IRET


; возврат из процедуры

KbProc ENDP

Init:


; начало нерезидентной части


MOV BX, 09H * 4
; вычисление смещения


XOR AX, AX


MOV ES, AX

; настройка ES на таблицу


MOV AX, ES:[BX]
; векторов прерываний

; сохранение значения старого вектора прерывания


MOV WORD PTR OLD_VEC, AX


MOV AX, ES:[BX + 2]


MOV WORD PTR OLD_VEC + 2, AX

; установка нового значения вектора прерывания


CLI


MOV ES:[BX], OFFSET KbProc


MOV ES:[BX + 2], CS


STI

; завершение программы и загрузка резидентной части


MOV DX, (Init – Start + 10FH) / 16


MOV AH, 31H


MOV AL, 0


INT 21H

CODE ENDS


END Start

Перед установкой нового значения вектора прерывания нужно запретить немаскируемые прерывания. Дело в том, что во время установки значения вектора прерывания может возникнуть сигнал прерывания и процессор попытается вызвать обработчик по вектору, значение которого ещё не до конца установлено. Это может повлечь за собой непредсказуемые результаты. Поэтому перед установкой значения вектора прерывания всегда нужно ставить команду запрета маскируемых прерываний CLI. По окончании установки значения вектора прерывания нужно разрешить маскируемые прерывания командой STI.

В этом примере нужно обратить внимание на следующие строки:





PUSHF


CALL CS:OLD_VEC

Дело в том, что вызвать старый обработчик через команду INT нельзя, поскольку его вектор прерывания указывает на наш обработчик. Поэтому нужно «смоделировать» работу команды INT, которая помимо занесения в стек значений регистров CS и IP заносит туда ещё и значение регистра флагов. Таким образом, перед вызовом дальней процедуры нужно занести в стек значение регистра флагов, чтобы при входе в старый обработчик стек имел такую же структуру, как если бы сначала вызывался старый обработчик при обработке процессором прерывания от клавиатуры.

Пример резидентной программы типа COM с переключением стеков и настройкой регистра DS на сегмент команд:
CODE SEGMENT


ASSUME CS:CODE, DS:CODE


ORG 100H

Start:


; начало резидентной части


JMP Init


OLD_VEC DD ?
; значение старого вектора

KbProc PROC

; процедура обработки прерывания


PUSHF


CALL CS:OLD_VEC
; вызов старого обработчика

MOV CS:OLD_AX, AX
; сохранение AX
MOV CS:OLD_SS, SS
; сохранение SS

MOV CS:OLD_SP, SP
; сохранение SP

MOV AX, CS

MOV SS, AX

; переключение стека

MOV SP, OFFSET Stac + 128



PUSH DS


; сохранение DS


PUSH CS


POP DS


; настройка DS на CS


POP DS


; восстановление DS

MOV AX, CS:OLD_SS

MOV SS, AX

; восстановление стека

MOV SP, CS:OLD_SP

MOV AX, CS:OLD_AX
; восстановление AX


IRET


; возврат из процедуры

KbProc ENDP

OLD_SS DW ?

OLD_SP DW ?

OLD_AX DW ?

Stac DB 128 DUP (?)

Init:


; начало нерезидентной части


MOV BX, 09H * 4
; вычисление смещения


XOR AX, AX


MOV ES, AX

; настройка ES на таблицу


MOV AX, ES:[BX]
; векторов прерываний

; сохранение значения старого вектора прерывания


MOV WORD PTR OLD_VEC, AX


MOV AX, ES:[BX + 2]


MOV WORD PTR OLD_VEC + 2, AX

; установка нового значения вектора прерывания


CLI


MOV ES:[BX], OFFSET KbProc


MOV ES:[BX + 2], CS


STI

; завершение программы и загрузка резидентной части


MOV DX, (Init – Start + 10FH) / 16


MOV AH, 31H


MOV AL, 0


INT 21H

CODE ENDS


END Start

Последний пример будет использоваться в дальнейшем для добавления к программе типа COM других элементов структуры резидентной программы.
Примеры простейших резидентных программы типа EXE


Следующий пример демонстрирует программу типа EXE с раздельными программными сегментами для различных элементов резидентной и нерезидентной частей. Действие программы аналогично первому примеру резидентной COM программы – вызывается старый обработчик прерывания от клавиатуры, переключения стеков не происходит.

Data SEGMENT


OLD_VEC DD ?

Data ENDS

Resident SEGMENT


ASSUME CS:Resident, DS:Data

KbProc PROC


PUSH DS


PUSH AX


MOV AX, Data


MOV DS, AX


PUSHF


CALL OLD_VEC


POP AX


POP DS


IRET

KbProc ENDP

Resident ENDS

Code SEGMENT PARA


ASSUME CS:Code, DS:Data

Start:


MOV AX, Data


MOV DS, AX


MOV BX, 09H * 4
; вычисление смещения


XOR AX, AX


MOV ES, AX

; настройка ES на таблицу


MOV AX, ES:[BX]
; векторов прерываний

; сохранение значения старого вектора прерывания


MOV WORD PTR OLD_VEC, AX


MOV AX, ES:[BX + 2]


MOV WORD PTR OLD_VEC + 2, AX

; установка нового значения вектора прерывания


CLI


MOV WORD PTR ES:[BX], OFFSET KbProc


MOV WORD PTR ES:[BX + 2], Resident


STI

; завершение программы и загрузка резидентной части


MOV DX, Code


SUB DX, Data


ADD DX, 10H


MOV AH, 31H


MOV AL, 0


INT 21H

Code ENDS

Stac SEGMENT PARA STACK


DB 128 DUP (?)

Stac ENDS


END Start

Конструкции «WORD PTR» в разделе изменения значения вектора прерывания необходимо указать, поскольку смещение процедуры KbProc может быть меньше 256 и тогда значение этого выражения можно записать как в байт, так и в слово; в следующей строке значение выражения Resident неизвестно на этапе трансляции и для него тоже необходимо указать тип.

Следующий пример дополняет предыдущий переключением стека в резидентной части.

Data SEGMENT


DB 32 DUP 0

Data ENDS

Stac SEGMENT PARA


DB 128 DUP (?)

Stac ENDS

Resident SEGMENT


ASSUME CS:Resident, DS:Data

KbProc PROC


PUSHF


CALL CS:OLD_VEC

MOV CS:OLD_AX, AX
; сохранение AX
MOV CS:OLD_SS, SS
; сохранение SS

MOV CS:OLD_SP, SP
; сохранение SP

MOV AX, Stac

MOV SS, AX

; переключение стека

MOV SP, 128



PUSH DS


; сохранение DS


MOV AX, Data


MOV DS, AX

; настройка DS на сегмент данных

POP DS


; восстановление DS

MOV AX, CS:OLD_SS

MOV SS, AX

; восстановление стека

MOV SP, CS:OLD_SP

MOV AX, CS:OLD_AX
; восстановление AX


IRET


OLD_SS DW ?


OLD_SP DW ?


OLD_AX DW ?

KbProc ENDP


OLD_VEC DD ?

Resident ENDS

Code SEGMENT PARA


ASSUME CS:Code, DS:Data

Start:


MOV AX, Data


MOV DS, AX


MOV BX, 09H * 4
; вычисление смещения


XOR AX, AX


MOV ES, AX

; настройка ES на таблицу


MOV AX, ES:[BX]
; векторов прерываний

; сохранение значения старого вектора прерывания


MOV WORD PTR OLD_VEC, AX


MOV AX, ES:[BX + 2]


MOV WORD PTR OLD_VEC + 2, AX

; установка нового значения вектора прерывания


CLI


MOV WORD PTR ES:[BX], OFFSET KbProc


MOV WORD PTR ES:[BX + 2], Resident


STI

; завершение программы и загрузка резидентной части


MOV DX, Code


SUB DX, Data


ADD DX, 10H


MOV AH, 31H


MOV AL, 0


INT 21H

Code ENDS

Stac2 SEGMENT PARA STACK


DB 128 DUP (?)

Stac2 ENDS


END Start

Последний пример будет использоваться в дальнейшем для добавления к программе типа EXE других элементов структуры резидентной программы.
Вызов старого обработчика прерывания


Вызов старого обработчика прерывания может выполняться как перед основным кодом резидентной части, так и после него. Конкретное расположение вызова обработчика зависит от того, следует ли выполнить какие-либо действия до того, как старый обработчик выполнит стандартные действия или нет. Если старый обработчик вызывается до основного кода резидентной части, то этот вызов выполняется так, как было показано в предыдущих примерах. Если необходимо вызвать старый обработчик после кода резидентной части, то это можно выполнить либо таким же способом, либо используя команду JMP, как показано в следующем примере:

OLD_VEC DD ?

. . .

MyProc PROC


. . .


JMP CS:OLD_VEC

MyProc ENDP

Следует отметить, что такой вызов должен выполняться в самом конце процедуры обработки прерывания, так как  управление на код, расположенный за этим вызовом, уже не вернётся. При таком вызове состояние стека должно быть таким же, как и при входе в процедуру обработки прерывания.

Связь с резидентной частью


При написании процедуры проверки на повторную загрузку возникает необходимость каким-либо образом узнать, загружена ли уже в памяти резидентная часть программы или нет. При написании процедуры выгрузки резидентной части программы из памяти, резидентной части требуется как-то передать команду на выгрузку из памяти. Во всех этих случаях необходимо организовать связь нерезидентной части программы с уже загруженной в память копией резидентной части программы. Связь можно организовавать, применив какое-либо прерывание. При этом перехватывается вектор этого прерывания и резидентная часть содержит процедуру его обработки. Тогда нерезидентная часть вызывает это прерывание, обработка которого попадает к резидентной части, а резидентная часть должна во-первых определить, что её вызвала «своя» программа, а во-вторых дать знать «своей» программе, что резидентная часть уже загружена. Также возникает вопрос, какой номер прерывания выбрать для организации связи. В MS-DOS существует специальное прерывание 2FH, которое называется мультиплексным прерыванием и служит для связи процессов. Для того, чтобы резидентная программа узнала о том, что её вызвала «своя» программа, нерезидентная часть программы должна записать в какие-либо регистры какие-либо уникальные значения, которые потом резидентная часть программы должна будет проверить. Если проверяемые резидентной частью значения совпадают с теми, которые должны быть, резидентная часть возвращает в каких-либо регистрах «ответ» тоже в виде каких-либо уникальных значений. Если значения не совпадают с требуемыми, то резидентная часть должна передать управление поцепочке старому обработчику, который также может вызвать ещё более старый обработчик и т. д. Таким образом организуется интерфейс нерезидентной части с резидентной. После того, как нерезидентная часть вызвала мультиплексное прерывание, она проверяет «ответ» резидентной части для проверки «своей» резидентной части. Если проверяемые нерезидентной частью значения совпали с требуемыми, то можно с уверенностью сказать, что резидентная копия программы уже загружена в память. 

Следующий пример демонстрирует программу, которая перехватывает вектор прерывания 2FH и проверяет, загружена ли резидентная часть программы в память или нет. Если резидентная часть уже загружена, программа завершается, если нет, то загружает резидентную часть и завершается.

CODE SEGMENT


ASSUME CS:CODE, DS:CODE


ORG 100H

Start:


JMP Init


OLD_2FH DD ?

New2FH PROC


CMP AX, 1234H

; проверка на “свою” программу


JNE Exit


CMP DX, 9876H


JNE Exit


MOV AX, 4321H

; запись в регистры ответа


MOV DX, 6789H


IRET



; возврат из процедуры

Exit:


JMP CS:OLD_2FH

; передача управления по цепочке

New2FH ENDP

Init:


MOV AX, 1234H

; запись в регистры 


MOV DX, 9876H

; уникальных значений


INT 2FH


; вызов мультиплексного






; прерывания


CMP AX, 4321H

; сравнение возвращённых


JNE Load


; значений (ответа)


CMP DX, 6789H

; с нужными


JNE Load


MOV AX, 4C00H

; завершение программы, если


INT 21H


; проверяемые значения совпали






; с требуемыми

Load:


MOV BX, 2FH * 4
; вычисление смещения


XOR AX, AX


MOV ES, AX

; настройка ES на таблицу


MOV AX, ES:[BX]
; векторов прерываний

; сохранение значения старого вектора прерывания


MOV WORD PTR OLD_2FH, AX


MOV AX, ES:[BX + 2]


MOV WORD PTR OLD_2FH + 2, AX

; установка нового значения вектора прерывания


CLI


MOV WORD PTR ES:[BX], OFFSET New2FH


MOV ES:[BX + 2], CS


STI

; завершение программы и загрузка резидентной части


MOV DX, (Init – Start + 10FH) / 16


MOV AH, 31H


MOV AL, 0


INT 21H

CODE ENDS


END Start

В примере функция DOS 4CH завершает программу. В регистре AL записывается код возврата (обычно 0).


Команды резидентной части программы можно передавать также, как и проверять её наличие – через прерывание 2FH с помощью регистров.

Процедура выгрузки резидентной части


Процедура выгрузки резидентной части программы должна располагаться в самой резидентной части, так как выгрузить резидентную часть из нерезидентной очень сложно. Поэтому резидентная часть сама выгружает себя из памяти по команде от нерезидентной части. Для выгрузки резидентной части требуется выполнить следующие действия:

1. восстановить все изменённые вектора прерываний, включая 2FH;

2. получить номер блока памяти, отведённого для окружения программы;

3. освободить блок памяти, занимаемый окружением программы;

4. освободить блок памяти, занимаемый самой программой.

Для получения блока памяти, занимаемого окружением программы, нужно прочитать значение слова, расположенного по смещению 2CH относительно префикса программного сегмента. Для программ типа COM это сделать легко, поскольку регистр CS настроен на начало префикса программного сегмента. Для программ типа EXE сегмент префикса программного сегмента можно получить, используя функцию 51H прерывания DOS 21H, которая в регистре BX возвращает значение сегмента префикса программного сегмента. При этом значение сегмента префикса программного сегмента должно быть сохранено в переменных резидентной программы, поскольку вызов функции 51H из резидентной части даст неверное значение сегмента префикса программного сегмента.


Для освобождения блока памяти можно использовать функцию DOS 49H, для которой в регистре ES указывается сегмент освобождаемого блока памяти.


Следующий пример демонстрирует процедуру обработки мультиплексного прерывания 2FH, которая освобождает занимаемую ей память по команде от нерезидентной части (программа типа COM).

OLD_2FH DD ?

. . .

New2FH PROC


CMP CX, 1234H

; проверка на “свою” программу


JNE Exit


CMP DX, 9876H


JNE Exit


CMP AL, 0


JNE M1


MOV CX, 4321H

; запись в регистры ответа


MOV DX, 6789H


JMP M2

M1:
CMP AL, 1


JNE M2

     CALL Unload

; вызов процедуры выгрузки

M2:
IRET



; возврат из процедуры

Exit:


JMP CS:OLD_2FH

; передача управления по цепочке

New2FH ENDP

Unload PROC NEAR

; процедура выгрузки


PUSH AX


; сохранение регистров


PUSH BX


PUSH ES


MOV BX, 2FH * 4
; вычисление смещения


XOR AX, AX


MOV ES, AX

; восстановление старого значения вектора прерывания


MOV AX, WORD PTR CS:OLD_2FH


MOV ES:[BX], AX


MOV AX, WORD PTR CS:OLD_2FH + 2


MOV ES:[BX + 2], AX

; получение сегмента блока памяти переменных окружения


MOV ES, CS:[2Ch]


MOV AH, 49H




INT 21H


; освобождение блока памяти


PUSH CS


POP ES


MOV AH, 49H




INT 21H


; освобождение блока памяти


POP ES


POP BX


POP AX


RET

Unload ENDP

Процедура New2FH использует значение регистра AL, передаваемое нерезидентной частью, в качестве команды для выполнения различных действий. При нулевом значении регистра AL процедура заносит в регистры CX и DX ответ нерезидентной части. При единичном щначении регистра AL вызывается процедура выгрузки резидентной части из памяти.


Следующий пример процедуры Unload демонстрирует выгрузку резидентной части из памяти для программы типа EXE.

OLD_2FH DD ?

Unload PROC NEAR

; процедура выгрузки


PUSH AX


; сохранение регистров


PUSH BX


PUSH ES


MOV BX, 2FH * 4
; вычисление смещения


XOR AX, AX


MOV ES, AX

; восстановление старого значения вектора прерывания


MOV AX, WORD PTR CS:OLD_2FH


MOV ES:[BX], AX


MOV AX, WORD PTR CS:OLD_2FH + 2


MOV ES:[BX + 2], AX

; получение сегмента блока памяти переменных окружения


MOV ES, CS:PSP


PUSH ES


MOV ES, ES:2CH


MOV AH, 49H




INT 21H


; освобождение блока памяти


POP ES


MOV AH, 49H




INT 21H


; освобождение блока памяти


POP ES


POP BX


POP AX


RET

PSP DW ?

Unload ENDP

Процедура обработки ключей командной строки


Все символы, которые были введены с клавиатуры после имени программы перед её запуском из операционной системы, хранятся в префиксе программного сегмента начиная со смещения 81H. По смещению 80H хранится байт, значение которого представляет длину строки введённых символов.


Процедура обработки ключей командной строки сначала ищет признак начала ключа, например, это может быть символ ‘/’ или ‘-’. Затем процедура проверяет символы после начала ключа и до следующего символа пробела или начала ключа. Если эти символы представляет собой ключ программы, то процедура обработки ключей переходит к выполнению необходимых действий. Если эти символы не являются ключом программы, выдаётся сообщение об ошибке. Можно при отсутствии параметров программы, когда значение байта по смещению 80H равно 0, выдавать информацию о программе и её ключах.

Пример резидентной программы


В заключение, далее приводится пример полностью законченной резидентной программы типа EXE, в которой используются все элементы обобщённой структуры резидентной программы. Программа перехватывает прерывание таймера и выводит в правом верхнем углу экрана значение переменной от 0 до 9, которое увеличивается с приходом каждого прерывания от таймера.
STACK_SIZE EQU 128

; размер стека 128 байт

Data1 SEGMENT


; сегмент данных резидентной части

        VAR DB 0


; переменная для вывода на экран

Data1 ENDS

Stack1 SEGMENT PARA

; сегмент стека резидентной части

        DB STACK_SIZE DUP (?)
; область памяти под стек

Stack1 ENDS

Resident SEGMENT


; сегмент команд резидентной части

        ASSUME CS:Resident, DS:Data1

Timer PROC


; процедура обработки прерывания таймера

        PUSHF


; создание в стеке структуры для IRET

        CALL CS:OLD_TIMER
; вызов старого обработчика

        MOV CS:OLD_AX, AX
; сохранение AX

        MOV CS:OLD_SS, SS
; сохранение SS

        MOV CS:OLD_SP, SP
; сохранение SP

        MOV AX, Stack1

; в AX – сегмент стека

        MOV SS, AX

; переключение стека

        MOV SP, STACK_SIZE

        PUSH DS


; сохранение DS

        PUSH ES


; сохранение ES

        PUSH BX


; сохранение BX


        MOV AX, Data1

; в AX – сегмент данных

        MOV DS, AX

; настройка DS на сегмент данных

        MOV AL, VAR

; загрузка в AL значения VAR

        ADD AL, '0'

; преобразование в ASCII-код цифры

        MOV AH, 1FH

; атрибут символа

        MOV BX, 0B800H

; настройка ES

        MOV ES, BX

; на сегмент видеопамяти

        MOV WORD PTR ES:158, AX
; запись символа в видеопамять

        INC VAR


; увеличение значения переменной

        CMP VAR, 10

; если меньше 10,

        JB M9


; то переход далее

        MOV VAR, 0

; иначе – обнуление значения

M9:     POP BX


; восстановление BX

        POP ES


; восстановление ES

        POP DS


; восстановление DS

        MOV AX, CS:OLD_SS
; в AX – старое значение SS

        MOV SS, AX

; переключение стека

        MOV SP, CS:OLD_SP

        MOV AX, CS:OLD_AX
; восстановление AX

        IRET


; возврат из процедуры

Timer ENDP

New2FH PROC

; процедура обработки мультиплексного прерывания

        CMP CX, 1234H

; проверка

        JNE Exit


; на «свою»

        CMP DX, 9876H

; программу

        JNE Exit

        CMP AL, 1


; если AL=1
        JE M1


; переход на выгрузку программы

        MOV CX, 4321H

; запись в регистры

        MOV DX, 6789H

; уникальных значений

        IRET


; возврат из процедуры

M1:     CALL Unload

; вызов процедуры выгрузки

        IRET


; возврат из процедуры

Exit:   JMP CS:OLD_2FH

; переход на старый обработчик

New2FH ENDP

        ASSUME DS:Resident

Unload PROC NEAR


; процедура выгрузки

        PUSH AX


; сохранение AX

        PUSH BX


; сохранение BX

        PUSH DS


; сохранение DS

        PUSH ES


; сохранение ES


        MOV AX, Resident
; в AX – сегмент команд

        MOV DS, AX

; настройка DS на сегмент команд

        MOV BX, 1CH * 4

; вычисление смещения

        XOR AX, AX

; настройка ES на начало

        MOV ES, AX

; таблицы векторов прерываний

; восстановление старых значений векторов прерываний

        MOV AX, WORD PTR OLD_TIMER

        MOV ES:[BX], AX

        MOV AX, WORD PTR OLD_TIMER + 2

        MOV ES:[BX + 2], AX

        MOV BX, 2FH * 4

        MOV AX, WORD PTR OLD_2FH

        MOV ES:[BX], AX

        MOV AX, WORD PTR OLD_2FH + 2

        MOV ES:[BX + 2], AX

; настройка ES на префикс программного сегмента

        MOV ES, PSP



        PUSH ES


; сохранение ES

        MOV ES, ES:2CH

; в ES – сегмент окружения программы

        MOV AH, 49H

; функция освобождения блока памяти

        INT 21H


; вызов DOS

        POP ES


; в ES – сегмент блока памяти программы

        MOV AH, 49H

; функция освобождения блока памяти

        INT 21H


; вызов DOS

        POP ES


; восстановление ES

        POP DS


; восстановление DS

        POP BX


; восстановление BX

        POP AX


; восстановление AX

        RET



; возврат из процедуры

Unload ENDP

        OLD_AX DW ?

; старое значение AX

        OLD_SS DW ?

; старое значение SS

        OLD_SP DW ?

; старое значение SP

        PSP DW ?


; значение сегмента префикса

; старое значение вектора прерывания от таймера

        OLD_TIMER DD ?



; старое значение вектора мультиплексного прерывания

        OLD_2FH DD ?

Resident ENDS

Code SEGMENT


; сегмент команд нерезидентной части

        ASSUME CS:Code, DS:Data2

Start:  MOV AX, Data2

; в AX – сегмент данных

        MOV DS, AX

; настройка DS на сегмент данных

; вызов процедуры обработки командной строки     

        CALL CmdLine



        CMP BX, 0


; проверка возвращённого значения

        JNE M2

        MOV DX, OFFSET Msg1
; смещение сообщения о программе

        JMP EndMsg

; переход на выдачу сообщения

M2:     MOV CX, 1234H

; запись в регистры

        MOV DX, 9876H

; уникальных значений

        MOV AL, 0


; команда резидентной части

        INT 2FH


; вызов мультиплексного прерывания

        CMP CX, 4321H

; проверка значений,

        JNE NoProg

; возвращаемых

        CMP DX, 6789H

; резидентной

        JNE NoProg

; частью

        CMP BX, 2


; если BX<>2, то

        JNE M8


; переход на выдачу сообщения

        CALL Unload2

; вызов процедуры выгрузки

        MOV DX, OFFSET Msg3
; смещение сообщения о выгрузке

        JMP EndMsg

; переход на выдачу сообщения

M8:     MOV DX, OFFSET Msg4
; смещение сообщения об ошибке

        JMP EndMsg

; переход на выдачу сообщения

NoProg: CMP BX, 1


; если BX=1, то переход на

        JE Load


; загрузку резидентной части

        MOV DX, OFFSET Msg2
; смещение сообщения об ошибке

EndMsg: MOV AH, 09H

; функция вывода строки

        INT 21H


; вызов DOS

        MOV AX, 4C00H

; функция завершения программы

        INT 21H


; вызов DOS
        ASSUME DS:Resident

Load PROC



; процедура загрузки резидентной части

        PUSH DS


; сохранение DS

; в AX – сегмент команд резидентной части
        MOV AX, Resident


; настройка DS на сегмент команд резидентной части

        MOV DS, AX



        XOR AX, AX

; настройка ES на начало

        MOV ES, AX

; таблицы векторов прерываний

        CLI



; запрет маскируемых прерываний

        MOV BX, 1CH * 4

; вычисление смещения

; установка нового значения вектора прерывания таймера

        MOV AX, ES:[BX]

        MOV WORD PTR OLD_TIMER, AX

        MOV AX, ES:[BX + 2]

        MOV WORD PTR OLD_TIMER + 2, AX

        MOV AX, OFFSET Timer

        MOV ES:[BX], AX

        MOV AX, Resident

        MOV ES:[BX + 2], AX

        MOV BX, 2FH * 4

; вычисление смещения

; установка нового значения вектора мультиплексного прерывания

        MOV AX, ES:[BX]

        MOV WORD PTR OLD_2FH, AX

        MOV AX, ES:[BX + 2]

        MOV WORD PTR OLD_2FH + 2, AX

        MOV AX, OFFSET New2FH

        MOV ES:[BX], AX

        MOV AX, Resident

        MOV ES:[BX + 2], AX

        STI



; разрешение маскируемых прерываний

        POP DS


; восстановление DS

        MOV DX, OFFSET Msg5
; смещение сообщения о загрузке

        MOV AH, 09H

; функция вывода строки

        INT 21H


; вызов DOS

        MOV DX, Code

; вычисление размера

        SUB DX, Data1

; резидентной части

        ADD DX, 10H

; в параграфах

; функция сохранения резидентной программы в памяти

        MOV AX, 3100H



        INT 21H


; вызов DOS

Load ENDP

Unload2 PROC


; процедура передачи команды выгрузки

        MOV CX, 1234H

; запись в регистры

        MOV DX, 9876H

; уникальных значений

        MOV AL, 1


; команда на выгрузку

        INT 2FH


; вызов мультиплексного прерывания

        RET



; возврат из процедуры

Unload2 ENDP

        ASSUME DS:Resident

CmdLine PROC


; процедура обработки командной строки

; функция получения сегмента префикса программного сегмента

        MOV AH, 51H



        INT 21H


; вызов DOS

        PUSH DS


; сохранение DS

; в AX – сегмент команд резидентной части
        MOV AX, Resident
; настройка DS на сегмент команд резидентной части

        MOV DS, AX

; запись в переменную резидентной части значение сегмента префикса

; программного сегмента

        MOV PSP, BX

; настройка DS на префикс программного сегмента

        MOV DS, BX

; настройка SI на начало командной строки

        MOV SI, 80H

        LODSB


; загрузка длины строки

        CMP AL, 0


; если строка не нулевой длины

        JNE M3


; переход на её обработку

        XOR BX, BX

; возвращаемое значение 0

        JMP M4


; переход на конец процедуры

M3:     LODSB


; загрузка очередного символа

        CMP AL, ' '

; пропуск пробелов

        JE M3


; в начале строки

        CMP AL, '/'

; проверка на признак ключа

        JE M5


; переход на обработку ключа

        XOR BX, BX

; возвращаемое значение 0

        JMP M4


; переход на конец процедуры

M5:     LODSB


; загрузка символа ключа

        CMP AL, 'i'

; проверки на возможные ключи

        JE M6

        CMP AL, 'I'

        JE M6

        CMP AL, 'u'
        JE M7

        CMP AL, 'U'

        JE M7

        XOR BX, BX

; возвращаемое значение 0

        JMP M4


; переход на конец процедуры

M6:     MOV BX, 1


; возвращаемое значение 1 - загрузка

        JMP M4


; переход на конец процедуры

M7:     MOV BX, 2


; возвращаемое значение 2 - выгрузка

M4:     POP DS


; восстановление DS

        RET



; возврат из процедуры

CmdLine ENDP

Code ENDS

Data2 SEGMENT


; сегмент данных нерезидентной части

        Msg1 DB 'TIMER.EXE </i | /u>$'

        Msg2 DB 'Program is not yet loaded!$'

        Msg3 DB 'Program successfully unloaded!$'

        Msg4 DB 'Program already loaded!$'

        Msg5 DB 'Program successfully loaded!$'

Data2 ENDS

Stack2 SEGMENT PARA STACK
; сегмент стека нерезидентной части

        DB 128 DUP (?)

; область памяти под стек

Stack2 ENDS

        END Start


; точка входа


Прерывание таймера 1CH вызывается обработчиком аппаратного прерывания таймера 08H, поэтому посылать в контроллер прерываний сигнал EOI в конце процедуры обработчика прерывания таймера Timer не нужно – это сделает системный обработчик.

В этом примере используется шесть программных сегментов – по на каждую часть программы (резидентную и нерезидентную). Каждая часть имеет сегмент команд, сегмент данных и сегмент стека. В нерезидентной части кроме своего сегмента данных в процедуре Load используется сегмент команд резидентной части Resident для сохранения старых значений векторов прерываний 2FH (мультиплексное прерывание) и 1CH (прерывание таймера) в переменных резидентной части. Эти переменные не вынесены в сегмент данных резидентной части, так как требуется вызывать старый обработчик за пределами группы команд переключения стека, а в резидентной части стек прерванной программы не используется. В результате нет возможности сохранить значения сегментных регистров для настройки на сегмент Data1 и приходится переменные со значениями векторов хранить в сегменте команд резидентной части.


Процедура Load нерезидентной части выполняет следующие действия:

1. сохраняет в переменных резидентной части старые значения векторов прерываний;

2. устанавливает в таблице векторов прерываний новые значения;

3. загружает в память резидентную часть и завершает нерезидентную часть с выдачей сообщения об успешной загрузке.

Следует обратить внимание на то, как вызывается процедура Load. Поскольку эта процедура завершает выполнение нерезидентной части, она не возвращает управление вызвавшей её программе, поэтому для вызова этой процедуры используется команда безусловного перехода JMP.


Процедура Unload2 передаёт резидентной части команду на выгрузку. Вслед за вызовом этой процедуры выводится сообщение об успешной выгрузке резидентной части программы из памяти.


Процедура CmdLine выполняет разбор командной строки и возвращает в  регистре BX следующие значения:

1. 0, если ключ командной строки неверный или отсутствуют параметры программы;

2. 1, если в командной строке указан ключ загрузки программы в память;

3. 2, если в командной строке указан ключ выгрузки программы из памяти.


Процедура Unload резидентной части выполняет выгрузку резидентной части программы типа EXE из памяти, как это было описано выше.


Процедура обработки прерывания таймера Timer выполняет переключение стека и выводит на экран значение переменной VAR. Для вывода на экран используется запись в видеопамять. Сегмент видеопамяти для текстового режима 80 x 25 имеет значение B800H. Это значение помещается в регистр ES и затем по смещению 158H, что соответствует верхнему правому углу экрана, выводится символ, представляющий значение переменной VAR, и атрибут 1FH – белый текст на синем фоне.

Процедура New2FH аналогична рассмотренной выше и выполняет команды, поступившие от нерезидентной части – возвращает ответ или запускает процедуру выгрузки Unload.


Для вывода сообщения на экран используется функция DOS 09H. Для выполнения этой функции в регистр AH необходимо занести значение 09H, а в регистр DX – смещение строки выводимых символов. Строка символов должна заканчиваться символом ‘$’, который на экран не выводится, но служит для DOS признаком конца строки. После занесения в регистры необходимых значений вызывается прерывание DOS 21H.

Общие рекомендации по написанию резидентных программ

1. При обращении резидентной части к данным, находящимся в другом сегменте, следует инициализировать сегментный регистр, при помощи которого резидентная часть будет обращаться к другому сегменту.

2. Резидентная часть обязательно должна вызывать старый обработчик прерывания, если он есть. Обычно все системные устройства, такие как клавиатура, таймер и др., имеют свои системные обработчики. При этом все регистры (кроме, конечно CS и IP) должны иметь то же значение, что и при входе в обработчик прерывания резидентной части. Стек также должен иметь исходное состояние.

3. Из процедур обработки аппаратных прерываний нельзя вызывать прерывания DOS и не рекомендуется вызывать другие прерывания. Это связано с тем, что во время возникновения аппаратного прерывания может выполняться какая-либо функция DOS, а вложенные вызовы функций DOS запрещены.

4. При обработке аппаратных прерываний следует обращать внимание на необходимость посылки сигнала EOI в конце процедуры обработки прерывания, если не вызывается старый обработчик. Если старый обработчик вызывается, то он выполняет всю необходимую работу с аппаратурой.

5. Следует обращать внимание на правильность вычисления размера резидентной части перед вызовом функции DOS 31H. Неверное вычисление размера резидентной части может привести к тому, что часть кода резидентной части не останется в памяти и это приведёт к непредсказуемым результатам.

6. При написании программ, которые используют мультиплексное прерывание 2FH следует обращать внимание на то, что некоторые программы-оболочки, например Dos Navigator, перехватывают прерывание 2FH и не передают управление старым обработчикам прерывания по цепочке. Это вызывает нарушение связи нерезидентной и резидентной частей программы. Для отладки такой программы можно запустить интерпретатор COMMAND.COM и уже из него загружать программу.

